Asymptotic analysis of Airy Eqn 07-23-16

N. T. Gladd

Initialization: Be sure the files *NTGStylesheet2.nb* and *NTGUtilityFunctions.m* is are in the same directory as that from which this notebook was loaded. Then execute the cell immediately below by mousing left on the cell bar to the right of that cell and then typing “shift” + “enter”. Respond “Yes” in response to the query to evaluate initialization cells.

In[5]:=
```
SetDirectory[NotebookDirectory[]];
(* set directory where source files are located *)
SetOptions[EvaluationNotebook[], (* load the StyleSheet *)
 StyleDefinitions → Get["NTGStylesheet2.nb"]];
Get["NTGUtilityFunctions.m"]; (* Load utilities package *)
```

Original notebook *Asymptotic Airy Eqn 10-10-15*

Purpose

I work through an asymptotic analysis of the Airy equation. This is a classical problem and there are many references.

Airy’s equation was analyzed by George Stokes in 1857. His struggles with it were even the amusing subject of one of his love letters to his bride-to-be.

“When the cat’s away the mice may play. You are the cat and I am the poor little mouse. I have been doing what I guess you won’t let me do when we are married, sitting up till 3 o’clock in the morning fighting hard against a mathematical difficulty. Some years ago I attacked an integral of Airy’s, and after a severe trial reduced it to a readily calculable form. But there was one difficulty about it which, though I tried till I almost made myself ill, I could not get over, and at last I had to give it up and profess myself unable to master it. I took it up again a few days ago, and after a two or three days’ fight, the last of which I sat up till 3, I at last mastered it. I don’t say you won’t let me work at such things, but you will keep me to more regular hours. A little out of the way now and then does not signify, but there should not be too much of it. It is not the mere sitting up but the hard thinking combined with it …….”

I Analysis

Airy’s equation is

$$\frac{d^2 f(x)}{d x^2} + x f(x) = 0$$

IA Obtaining an integral representation

The function satisfying this differential equation can be expressed as a contour integral and useful asymptotic properties can be deduced using the method of steepest descent. Although the transformation into integral form can be quickly accomplished by a hand calculation, I persist in performing the required operations with Mathematica.

To work in the standard notation I use for such problems I write the differential equation using k as the independent variable

In[7]:=
```
w1A[1] = D[f[k], {k, 2}] + k f[k] == 0
```
Out[7]=
```
k f[k] + f''[k] == 0
```

Assume that the integral has the form of a contour integral

$$f(k) = \int_C \mathrm{dz}\, e^{k z} \mathcal{F}(z)$$

I represent this form with the structure

In[8]:=
```
w1A[2] = Int[Exp[k z] 𝓕[z], C]
```
Out[8]=
```
Int[ⅇ^(k z) 𝓕[z], C]
```

Then

In[9]:=
```
w1A[3] =
  w1A[1] /. {f[k] → Int[Exp[k z] 𝓕[z], C], f''[k] → Int[D[Exp[k z], {k, 2}] 𝓕[z], C]}
```
Out[9]=
```
k Int[ⅇ^(k z) 𝓕[z], C] + Int[ⅇ^(k z) z^2 𝓕[z], C] == 0
```

Integrate the first term by parts

In[10]:=
```
w1A[4] = w1A[3] /.
   Int[ⅇ^(k z) ℱ[z], C] → BoundaryTerm[ⅇ^(k z) ℱ[z], C] - Int[ⅇ^(k z)/k D[ℱ[z], z], C] // Expand
```

Out[10]=
$$k\,\text{BoundaryTerm}\left[e^{k z}\mathcal{F}[z], C\right] + \text{Int}\left[e^{k z} z^2 \mathcal{F}[z], C\right] - k\,\text{Int}\left[\frac{e^{k z}\mathcal{F}'[z]}{k}, C\right] == 0$$

where the “BoundaryTerm” term represents the value of the first argument at the end points of the contour C.

Assume that the contour is C is chosen so that the boundary term vanishes.

In[11]:=
```
w1A[5] = w1A[4] /. BoundaryTerm[arg_, con_] → 0 /.
   a_. Int[b_ c_, d_] /; FreeQ[b, z] → a b Int[c, d]
```

Out[11]=
$$\text{Int}\left[e^{k z} z^2 \mathcal{F}[z], C\right] - \text{Int}\left[e^{k z}\mathcal{F}'[z], C\right] == 0$$

I use a rewrite rule to combine the two terms

In[12]:=
```
w1A[6] = w1A[5] /. c1_. Int[a_, con_] + c2_. Int[b_, con_] → Int[c1 a + c2 b, con]
```

Out[12]=
$$\text{Int}\left[e^{k z} z^2 \mathcal{F}[z] - e^{k z}\mathcal{F}'[z], C\right] == 0$$

The integral only vanishes if the integrand vanishes

In[13]:=
```
w1A[7] = w1A[6][[1, 1]] == 0
```

Out[13]=
$$e^{k z} z^2 \mathcal{F}[z] - e^{k z}\mathcal{F}'[z] == 0$$

Solve this ode

In[14]:=
```
w1A[8] = DSolve[w1A[7], ℱ[z], z][[1, 1]] /. C[1] → κ // ER
```

Out[14]=
$$\mathcal{F}[z] \to e^{\frac{z^3}{3}}\kappa$$

where κ is a constant of integration. Thus,

In[15]:=
```
w1A[9] = w1A[2] /. w1A[8] /. Int[a_ b_, c_] /; FreeQ[a, z] → a Int[b, c]
```

Out[15]=
$$\kappa\,\text{Int}\left[e^{k z+\frac{z^3}{3}}, C\right]$$

or, the explicit integral representation of f(k) is

$$f(k) = \kappa \int_C dz\, e^{k z+\frac{z^3}{3}}$$

provided that the contour C is chosen so that the integrand vanishes at the end points of C.

1B Choice of contour

For z >> 1 the $e^{\frac{z^3}{3}}$ term dominates the integrand



Note: Such figures are generated in an Appendix

Choose a polar representation of z

In[17]:=
```
w1B[1] = Exp[-z^3/3] /. z → r Exp[I θ]
```
Out[17]= $e^{-\frac{1}{3} e^{3 i \theta} r^3}$

The condition on *C* requires Re$\left[e^{z^3/3}\right]$ → 0 as r →∞ and the contour plot makes it clear that this can only occur in three wedge shaped segments of the complex plane.

In[18]:=
```
w1B[2] = ComplexExpand[w1B[1]]
```
Out[18]= $e^{-\frac{1}{3} r^3 \text{Cos}[3\theta]} \text{Cos}\left[\frac{1}{3} r^3 \text{Sin}[3\theta]\right] - i\, e^{-\frac{1}{3} r^3 \text{Cos}[3\theta]} \text{Sin}\left[\frac{1}{3} r^3 \text{Sin}[3\theta]\right]$

The real part is

In[19]:=
```
w1B[3] = Re[w1B[2][[1]]] // Simplify[#, {r ∈ Reals, θ ∈ Reals}] &
```
Out[19]= $e^{-\frac{1}{3} r^3 \text{Cos}[3\theta]} \text{Cos}\left[\frac{1}{3} r^3 \text{Sin}[3\theta]\right]$

The sign of the argument of the exponential term is controlled by *θ*

In[20]:=
```
RegionPlot[-1/3 r^3 Cos[3 θ] < 0, {r, 0, 1}, {θ, -π/3, 2 π - π/3},
 FrameTicks → {Automatic, Table[π/6 i, {i, -1, 12}]}, FrameLabel → {"r", "θ"}]
```

Out[20]=



The allowed ranges of θ (blue regions) where the integrand becomes vanishing small as r → ∞ are

$$-\frac{\pi}{6} < \theta < \frac{\pi}{6}, \quad \frac{\pi}{2} < \theta < \frac{5\pi}{6}, \quad \frac{7\pi}{6} < \theta < \frac{3\pi}{2}$$

To satisfy the boundary condition, a contour must start at r = ∞ in one of the regions where the integrand is vanishingly small and end at r = ∞ in another such region. The case where a contour starts and ends in the same range is not allowed since the contour would constitute a closed path and, with no poles contained within the contour, the integral would have zero value. There are two independent solutions of the Airy equation, the Airy function and the BAiry function. By convention the contours associated with these solutions are



2 Asymptotic expansion of Ai(λ) for $\lambda > 0$

The integral representation of the Airy function is

$$A_i(\lambda) = \frac{1}{2\pi i}\int_{C_1} \mathrm{dz}\,\exp\left(-\frac{z^3}{3} + \lambda z\right)$$

where the $\frac{1}{2\pi i}$ is the value of the normalization constant κ chosen by convention. The first step is to recast this integral into the standard form convenient for asymptotic analysis using steepest descent method.

$$\int_C \mathrm{dz}\, e^{k\,\rho(z)} = \int_C \mathrm{dz}\, e^{k\,(\phi(z)+\mathrm{i}\psi(z))}$$

In[22]:=
```
w2[1] = 1/(2 π I) Int[Exp[- s^3/3 + λ s] ds, C1]
```

Out[22]=
$$-\frac{\mathrm{i}\,\mathrm{Int}\left[\mathrm{ds}\, e^{-\frac{s^3}{3}+s\lambda}, C_1\right]}{2\pi}$$

Rescale the integration variable

In[23]:=
```
w2[2] = w2[1] /. s → s[z] /. ds → D[s[z], z]
```

Out[23]=
$$-\frac{\mathrm{i}\,\mathrm{Int}\left[e^{\lambda s[z]-\frac{s[z]^3}{3}}\, s'[z], C_1\right]}{2\pi}$$

In[24]:=
```
w2[3] = w2[2] /. s → ((√λ #) &)
```
Out[24]=
$$-\frac{i\,\mathrm{Int}\left[e^{z\,\lambda^{3/2}-\frac{1}{3}z^3\lambda^{3/2}}\sqrt{\lambda},\ C_1\right]}{2\pi}$$

A change of parameter casts this into standard form

In[25]:=
```
w2[4] = w2[3] /. λ → k^(2/3) //. Int[a_. b_, c_] /; FreeQ[a, z] → a Int[b, c]
```
Out[25]=
$$-\frac{i\,k^{1/3}\,\mathrm{Int}\left[e^{k\,z-\frac{k\,z^3}{3}},\ C_1\right]}{2\pi}$$

The ρ(z) for this problem is

In[26]:=
```
w2[5] = ρ[z] == (k z - k z^3/3)/k // ExpandAll
```
Out[26]=
$$\rho[z] == z-\frac{z^3}{3}$$

I implement the function *PropertiesOfρ* to calculate some quantities useful for a steepest descent analysis

In[27]:=
```
Clear[PropertiesOfρ];
PropertiesOfρ[ρ_] :=
  Module[{saddlePoints, ϕ, ψ, d2ρ, w},
   saddlePoints = Solve[D[ρ, z] == 0];
   w[1] = ρ /. z → x + I y // ComplexExpand;
   {ϕ, ψ} = w[1] /. ϕ_ + I ψ_ → {ϕ, ψ};
   d2ρ = D[ρ, {z, 2}];
   Association[
    {"ρ" → ρ, "saddlePoints" → saddlePoints, "ϕ" → ϕ, "ψ" → ψ, "d2ρ" → d2ρ}]];
```

In[29]:=
```
Aρ = PropertiesOfρ[w2[5][[2]]]
```
Out[29]=
$$\langle|\ \rho\to z-\frac{z^3}{3},\ \mathrm{saddlePoints}\to\{\{z\to-1\},\{z\to1\}\},$$
$$\phi\to x-\frac{x^3}{3}+x\,y^2,\ \psi\to y-x^2\,y+\frac{y^3}{3},\ \mathrm{d2}\rho\to-2\,z\ |\rangle$$

In[30]:=
```
Normal[Aρ] // ColumnForm
```

Out[30]=

$\rho \to z - \frac{z^3}{3}$

saddlePoints → {{z → −1}, {z → 1}}

$\phi \to x - \frac{x^3}{3} + x\, y^2$

$\psi \to y - x^2\, y + \frac{y^3}{3}$

$d2\rho \to -2\, z$

There are two saddle points at -1 and 1. I determine the constant ψ curves passing through the saddle points.

The value of ρ(-1) is

In[31]:=
```
w2[6] = Aρ["ρ"] /. z → -1
```

Out[31]=

$-\frac{2}{3}$

which is real so ψ(z = -1) = 0 at the saddle point. The equation describing the steepest descent curves passing through the saddle point z = -1 is

In[32]:=
```
w2[7] = Aρ["ψ"] == 0
```

Out[32]=

$y - x^2\, y + \frac{y^3}{3} == 0$

The constant ψ curves passing through the saddle point z = -1 are

In[33]:=
```
w2[8] = Solve[w2[7], y]
```

Out[33]=

$\left\{\{y \to 0\},\ \left\{y \to -\sqrt{3}\ \sqrt{-1 + x^2}\right\},\ \left\{y \to \sqrt{3}\ \sqrt{-1 + x^2}\right\}\right\}$

Similarly, the value of ρ(1) is

In[34]:=
```
w2[9] = Aρ["ρ"] /. z → 1
```

Out[34]=

$\frac{2}{3}$

In[35]:=
```
w2[10] = Aρ["ψ"] == 0
```

Out[35]=

$y - x^2\, y + \frac{y^3}{3} == 0$

which is the same equation as for saddle point 1

The next step is to examine the saddle points of the integrand and determine how to deform the contour C_1 so that will pass through a saddle point along curves of steepest descent.

Out[36]=



From the figure it is obvious that the path $\mathcal{D}_1$ is a steepest descent path. The objective is to deform C_1 so that it passes through the saddle point at z_{SP1} = -1 along the steepest descent path $\mathcal{D}_1$.

A strategy for distorting C_1 into $\mathcal{D}_1$ is to start at some point {0, -yMax} on C_1, move along $\mathcal{L}_1$ to the steepest descent path $\mathcal{D}_1$, follow $\mathcal{D}_1$ to y = yMax, then return along $\mathcal{L}_2$ to C_1. If the contributions along $\mathcal{L}_1$ and $\mathcal{L}_2$ are negligible as yMax → ∞, then the integration along $\mathcal{D}_1$ is equivalent to integration along C_1. This is allowed since there are no intervening poles.



Since $\mathcal{L}_1$ and $\mathcal{L}_2$ are arbitrarily deep in the zones where $e^{k\,\rho(z)}$ is tending to zero, their contributions are clearly small with respect to the contribution near the saddle point. In a more formal sense, it is straight-forward to calculate that the contributions along $\mathcal{L}_1$ and $\mathcal{L}_2$ are bounded and that those bounds tend to zero as $Y_{max} \to \infty$.

3 Asymptotic approximation for Ai(λ) for $\lambda > 0$

So far,

$$A_i(\lambda) = \frac{1}{2\pi i}\int_{C_1} dz \exp\left(-\frac{z^3}{3} + \lambda z\right) = \frac{i k^{1/3}}{2\pi i}\int_{C_1} dz\, e^{k\left(z-\frac{z^3}{3}\right)} \simeq \frac{i k^{1/3}}{2\pi i}\int_{\mathcal{D}_1} dz\, e^{k\left(z-\frac{z^3}{3}\right)}$$

Further, for the purpose of calculating the leading order asymptotic expression, note that the steepest descent contour $\mathcal{D}_1$ is approximated by the straight line z = -1 + i y in the vicinity of the saddle point

In[39]:=
```
w3[1] = w2[4] /. C1 → Dapprox
```
Out[39]=
$$-\frac{i\, k^{1/3}\, \text{Int}\left[e^{k z-\frac{k z^3}{3}}, \mathcal{D}_{approx}\right]}{2\pi}$$

In[40]:=
```
w3[2] = w3[1] /. e^(k z - k z^3/3) → e^(k ρ[z])
```
Out[40]=
$$-\frac{i\, k^{1/3}\, \text{Int}\left[e^{k \rho[z]}, \mathcal{D}_{approx}\right]}{2\pi}$$

Approximate ρ by a Taylor expansion about the saddle point

In[41]:=
```
w3[3] = w3[2] /. ρ[z] → Normal@Series[ρ[z], {z, z0, 2}]
```
Out[41]=
$$-\frac{1}{2\pi} i\, k^{1/3}\, \text{Int}\left[e^{k\left(\rho[z0]+(z-z0)\,\rho'[z0]+\frac{1}{2}(z-z0)^2\rho''[z0]\right)}, \mathcal{D}_{approx}\right]$$

In[42]:=
```
w3[4] = {w2[5], D[#, z] & /@ w2[5], D[#, {z, 2}] & /@ w2[5]} /. z → z0
```
Out[42]=
$$\left\{\rho[z0] == z0 - \frac{z0^3}{3},\ \rho'[z0] == 1 - z0^2,\ \rho''[z0] == -2\, z0\right\}$$

In[43]:=
```
w3[5] = w3[4] /. z0 → -1 // ER
```
Out[43]=
$$\left\{\rho[-1] \to -\frac{2}{3},\ \rho'[-1] \to 0,\ \rho''[-1] \to 2\right\}$$

In[44]:=
```
w3[6] = w3[3] /. z0 → -1 /. w3[5]
```
Out[44]=
$$-\frac{i\, k^{1/3}\, \text{Int}\left[e^{k\left(-\frac{2}{3}+(1+z)^2\right)}, \mathcal{D}_{approx}\right]}{2\pi}$$

Change variables in manner that takes into account the functional variation of the differential element

In[45]:=
```
w3[7] = w3[6] /. z → z[s] /. a_.Int[b_, c_] → a Int[b D[z[s], s], c]
```

Out[45]=

$$-\frac{1}{2\pi} i\, k^{1/3}\, \text{Int}\left[e^{k\left(-\frac{2}{3}+(1+z[s])^2\right)} z'[s], \mathcal{D}_{\text{approx}}\right]$$

Express the integrand in terms of the approximate contour $\mathcal{D}_{1\,\text{app}}$

In[46]:=
```
w3[8] = w3[7] /. z → ((-1 + I #) &) // ExpandAll
```

Out[46]=

$$-\frac{i\, k^{1/3}\, \text{Int}\left[i\, e^{-\frac{2k}{3}-k s^2}, \mathcal{D}_{\text{approx}}\right]}{2\pi}$$

Since the integrand is peaked about the saddle point (the motivating reason for changing the contour of integration), the contour of integration can be extended to ∞

In[47]:=
```
w3[9] = w3[8] /. 𝒟approx → {s, -∞, ∞}
```

Out[47]=

$$-\frac{i\, k^{1/3}\, \text{Int}\left[i\, e^{-\frac{2k}{3}-k s^2}, \{s, -\infty, \infty\}\right]}{2\pi}$$

Finally, I invoke Mathematica's Integration routine

In[48]:=
```
w3[10] = w3[9] /. Int → Integrate
```

Out[48]=

$$\text{ConditionalExpression}\left[\frac{e^{-2k/3}}{2\, k^{1/6}\sqrt{\pi}}, \text{Re}[k] > 0\right]$$

In[49]:=
```
w3[11] = Simplify[w3[10], Assumptions → Re[k] > 0]
```

Out[49]=

$$\frac{e^{-2k/3}}{2\, k^{1/6}\sqrt{\pi}}$$

Returning to the original parameter

In[50]:=
```
w3[12] = w3[11] /. k → λ^(3/2) // PowerExpand
```

Out[50]=

$$\frac{e^{-\frac{2\lambda^{3/2}}{3}}}{2\sqrt{\pi}\,\lambda^{1/4}}$$

Stating this result in standard asymptotic notation

$$A_i(\lambda) \sim \frac{e^{-\frac{2\lambda^{3/2}}{3}}}{2\sqrt{\pi}\,\sqrt[4]{\lambda}} \text{ as } (\lambda \to \infty)$$

I check this against Mathematica's implementation of the Airy function

In[51]:=

```
Module[{lab, F, FAsymptotic},
 F[λ_] := AiryAi[λ];
 FAsymptotic[λ_] := e^(-2 λ^(3/2)/3) / (2 Sqrt[π] λ^(1/4));

 lab = Stl@StringForm["Airy vs leading order asymptotic approximation"];
 Plot[{F[k], FAsymptotic[k]}, {k, 1, 5}, AxesLabel → {Stl["λ"], Stl["F"]},
  PlotLabel → lab, PlotStyle → {Black, Blue, Darker[Green, 0.5]},
  PlotLegends → Placed[{"AiryAi[λ]", "e^(-2 λ^(3/2)/3) / (2 Sqrt[π] λ^(1/4))"}, Right]]]
```

Out[51]=



4 Asymptotic expansion of Ai(λ) for $\lambda < 0$

For this case I write

$$A_i(\Lambda) = \frac{1}{2\pi i}\int_{C_1} dz \exp\left(-\frac{z^3}{3} - \Lambda z\right)$$

where $\Lambda = -\lambda$ and is a positive quantity. Note the sign change in the argument of the exponential.

In[52]:=

```
w4[1] = 1/(2 π I) Int[Exp[- s^3/3 - Λ s] ds, C1]
```

Out[52]=

$$-\frac{i\,\text{Int}\left[\text{ds}\, e^{-\frac{s^3}{3}-s\Lambda}, C_1\right]}{2\pi}$$

In[53]:=
```
w4[2] = w4[1] /. s → s[z] /. ds → D[s[z], z]
```

Out[53]=

$$-\frac{i\,\text{Int}\left[e^{-\Lambda s[z]-\frac{s[z]^3}{3}} s'[z], C_1\right]}{2\pi}$$

In[54]:=
```
w4[3] = w4[2] /. s → ((√Λ #) &)
```

Out[54]=

$$-\frac{i\,\text{Int}\left[e^{-z\Lambda^{3/2}-\frac{1}{3}z^3\Lambda^{3/2}}\sqrt{\Lambda}, C_1\right]}{2\pi}$$

In[55]:=
```
w4[4] = w4[3] /. Λ → k^(2/3) //. Int[a_.b_, c_] /; FreeQ[a, z] → a Int[b, c]
```

Out[55]=

$$-\frac{i\,k^{1/3}\,\text{Int}\left[e^{-k z-\frac{k z^3}{3}}, C_1\right]}{2\pi}$$

The ρ(z) for this problem is

In[56]:=
```
w4[5] = ρ[z] == (-k z - k z^3/3) / k // ExpandAll
```

Out[56]=

$$\rho[z] == -z - \frac{z^3}{3}$$

As before

In[57]:=
```
Aρ2 = PropertiesOfρ[w4[5][[2]]];
Normal[Aρ2] // ColumnForm
```

Out[58]=

$\rho \to -z - \frac{z^3}{3}$

saddlePoints → {{z → -i}, {z → i}}

$\phi \to -x - \frac{x^3}{3} + x y^2$

$\psi \to -y - x^2 y + \frac{y^3}{3}$

d2ρ → -2 z

In this case, there are two saddle points at -i and i. The value of ρ(-i) is

In[59]:=
```
w4[6] = Aρ2["ρ"] /. z → -I
```

Out[59]=

$$\frac{2\,i}{3}$$

The equation describing the steepest descent curves passing through the saddle point z = -i is

In[60]:=
```
w4[7] = Aρ2["ψ"] == 2/3
```

Out[60]=

$$-y - x^2 y + \frac{y^3}{3} == \frac{2}{3}$$

In this case it is easier to solve for x(y).

In[61]:=
```
w4[8] = Solve[w4[7], x]
```

Out[61]=

$$\left\{\left\{x \to -\frac{\sqrt{-2+y}\,(1+y)}{\sqrt{3}\,\sqrt{y}}\right\}, \left\{x \to \frac{\sqrt{-2+y}\,(1+y)}{\sqrt{3}\,\sqrt{y}}\right\}\right\}$$

Similarly, the value of ρ(i) is

In[62]:=
```
w4[9] = Aρ2["ρ"] /. z → I
```

Out[62]=

$$-\frac{2\,i}{3}$$

The equation describing the steepest descent curves passing through the saddle point z = i is

In[63]:=
```
w4[10] = Aρ2["ψ"] == -2/3
```

Out[63]=

$$-y - x^2 y + \frac{y^3}{3} == -\frac{2}{3}$$

In[64]:=
```
w4[11] = Solve[w4[10], x]
```

Out[64]=

$$\left\{\left\{x \to -\frac{(-1+y)\,\sqrt{2+y}}{\sqrt{3}\,\sqrt{y}}\right\}, \left\{x \to \frac{(-1+y)\,\sqrt{2+y}}{\sqrt{3}\,\sqrt{y}}\right\}\right\}$$

The steepest descent paths passing through the two saddle points at z = -i and z = i are



The strategy for distorting the contour C_1 is to pull the midpoint of the contour to the right along the positive axis and adjust it to coincide with $\mathcal{D}_1$ and $\mathcal{D}_2$. This is allowed since there are no intervening poles.

5 Asymptotic approximation for Ai(λ) for λ < 0

So far,

$$A_i(\Lambda) = \frac{1}{2\pi i}\int_{C_1} dz \exp\left(-\frac{z^3}{3} - \Lambda z\right) = \frac{i\,k^{1/3}}{2\pi i}\int_{C_1} dz\, e^{k\left(-z-\frac{z^3}{3}\right)} \simeq \frac{i\,k^{1/3}}{2\pi i}\int_{\mathcal{D}_1} dz\, e^{k\left(-z-\frac{z^3}{3}\right)} + \frac{i\,k^{1/3}}{2\pi i}\int_{\mathcal{D}_2} dz\, e^{k\left(-z-\frac{z^3}{3}\right)}$$

In this case, there are contributions from both saddle points.

5A Contribution from steepest descent curve $\mathcal{D}_1$ passing through saddle point z = -i

In[66]:=
```
w5A[1] = w4[4] /. C1 → D1
```

Out[66]=
$$-\frac{i\, k^{1/3}\, \mathrm{Int}\left[e^{-k z - \frac{k z^3}{3}}, \mathcal{D}_1\right]}{2\pi}$$

In[67]:=
```
w5A[2] = w5A[1] /. e^(-k z - k z^3/3) → e^(k ρ[z])
```

Out[67]=
$$-\frac{i\, k^{1/3}\, \mathrm{Int}\left[e^{k\rho[z]}, \mathcal{D}_1\right]}{2\pi}$$

Approximate ρ by a Taylor expansion about the saddle point

In[68]:=
```
w5A[3] = w5A[2] /. ρ[z] → Normal@Series[- z - z^3/3, {z, z0, 2}]
```

Out[68]=
$$-\frac{1}{2\pi} i\, k^{1/3}\, \text{Int}\left[e^{k\left(-z0-(z-z0)^2 z0-\frac{z0^3}{3}+(z-z0)\left(-1-z0^2\right)\right)}, \mathcal{D}_1\right]$$

In[69]:=
```
w5A[4] = {w4[5], D[#, z] & /@ w4[5], D[#, {z, 2}] & /@ w4[5]} /. z → z0
```

Out[69]=
$$\left\{\rho[z0] == -z0-\frac{z0^3}{3},\ \rho'[z0] == -1-z0^2,\ \rho''[z0] == -2\, z0\right\}$$

In[70]:=
```
w5A[5] = w5A[4] /. z0 → -I // ER
```

Out[70]=
$$\left\{\rho[-i] \to \frac{2\, i}{3},\ \rho'[-i] \to 0,\ \rho''[-i] \to 2\, i\right\}$$

In[71]:=
```
w5A[6] = w5A[3] /. z0 → -I /. w5A[5]
```

Out[71]=
$$-\frac{i\, k^{1/3}\, \text{Int}\left[e^{k\left(\frac{2 i}{3}+i\,(i+z)^2\right)}, \mathcal{D}_1\right]}{2\pi}$$

Change variables

In[72]:=
```
w5A[7] = w5A[6] /. z → z[s] /. a_.Int[b_, c_] → a Int[b D[z[s], s], c]
```

Out[72]=
$$-\frac{1}{2\pi} i\, k^{1/3}\, \text{Int}\left[e^{k\left(\frac{2 i}{3}+i\,(i+z[s])^2\right)} z'[s], \mathcal{D}_1\right]$$

The approximate functional form for $\mathcal{D}_1$ is z → x + i (x-1)   Handling the Mathematica representation of the equation for the curves of steepest descent requires some care. See Appendix A for detail

The straight line approximation for $\mathcal{D}_1$ in the vicinity of z_{SP1} is

In[73]:=
```
w5A[8] = w5A[7] /. z → ((# + I (# - 1)) &)  // ExpandAll
```

Out[73]=
$$-\frac{i\, k^{1/3}\, \text{Int}\left[(1+i)\, e^{\frac{2 i k}{3}-2 k s^2}, \mathcal{D}_1\right]}{2\pi}$$

Since this is a steepest descent path, the linear path may be extended to ∞

In[75]:=
```
w5A[9] = w5A[8] /. 𝒟1 → {s, -∞, ∞}
```

Out[75]=
$$-\frac{1}{2\pi} i\, k^{1/3}\, \text{Int}\left[(1+i)\, e^{\frac{2 i k}{3}-2 k s^2}, \{s, -\infty, \infty\}\right]$$

In[76]:=
```
w5A[10] = w5A[9] /. Int → Integrate
```

Out[76]=
$$\text{ConditionalExpression}\left[\frac{\left(\frac{1}{2}-\frac{i}{2}\right)e^{\frac{2 i k}{3}}}{k^{1/6}\sqrt{2\pi}}, \text{Re}[k]>0\right]$$

In[77]:=
```
w5A[11] = Simplify[w5A[10], Assumptions → Re[k] > 0]
```

Out[77]=
$$\frac{\left(\frac{1}{2}-\frac{i}{2}\right)e^{\frac{2 i k}{3}}}{k^{1/6}\sqrt{2\pi}}$$

In[78]:=
```
w5A[12] = w5A[11] /. (1/2 - i/2) -> PolarForm[(1/2 - i/2)]
```

Out[78]=
$$\frac{e^{\frac{2 i k}{3}-\frac{i\pi}{4}}}{2 k^{1/6}\sqrt{\pi}}$$

Returning to the original parameter

In[79]:=
```
w5A[13] = w5A[12] /. k → Λ^(3/2) // PowerExpand
```

Out[79]=
$$\frac{e^{-\frac{i\pi}{4}+\frac{2}{3} i \Lambda^{3/2}}}{2\sqrt{\pi}\,\Lambda^{1/4}}$$

5B Contribution from steepest descent curve $\mathcal{D}_2$ passing through saddle point z = i

In[80]:=
```
w5B[1] = w4[4] /. C1 → D2
```

Out[80]=
$$-\frac{i\, k^{1/3}\,\text{Int}\left[e^{-k z-\frac{k z^3}{3}}, \mathcal{D}_2\right]}{2\pi}$$

In[81]:=
```
w5B[2] = w5B[1] /. e^(-k z - k z^3/3) → e^(k ρ[z])
```

Out[81]=
$$-\frac{i\, k^{1/3}\,\text{Int}\left[e^{k\rho[z]}, \mathcal{D}_2\right]}{2\pi}$$

Approximate ρ by a Taylor expansion about the saddle point

In[82]:=
```
w5B[3] = w5B[2] /. ρ[z] → Normal@Series[- z - z^3/3, {z, z0, 2}]
```

Out[82]=
$$-\frac{1}{2\pi} i\, k^{1/3}\, \text{Int}\left[e^{k\left(-z0-(z-z0)^2 z0-\frac{z0^3}{3}+(z-z0)\left(-1-z0^2\right)\right)}, \mathcal{D}_2\right]$$

In[83]:=
```
w5B[4] = {w4[5], D[#, z] & /@ w4[5], D[#, {z, 2}] & /@ w4[5]} /. z → z0
```

Out[83]=
$$\left\{\rho[z0] == -z0 - \frac{z0^3}{3},\ \rho'[z0] == -1 - z0^2,\ \rho''[z0] == -2\, z0\right\}$$

In[84]:=
```
w5B[5] = w5B[4] /. z0 → I // ER
```

Out[84]=
$$\left\{\rho[i] \to -\frac{2 i}{3},\ \rho'[i] \to 0,\ \rho''[i] \to -2 i\right\}$$

In[85]:=
```
w5B[6] = w5B[3] /. z0 → I /. w5B[5]
```

Out[85]=
$$-\frac{i\, k^{1/3}\, \text{Int}\left[e^{k\left(-\frac{2 i}{3}-i(-i+z)^2\right)}, \mathcal{D}_2\right]}{2\pi}$$

Change variables

In[86]:=
```
w5B[7] = w5B[6] /. z → z[s] /. a_. Int[b_, c_] → a Int[b D[z[s], s], c]
```

Out[86]=
$$-\frac{1}{2\pi} i\, k^{1/3}\, \text{Int}\left[e^{k\left(-\frac{2 i}{3}-i(-i+z[s])^2\right)} z'[s], \mathcal{D}_2\right]$$

The straight line approximation for $\mathcal{D}_2$ in the vicinity of z_{SP2} is

In[87]:=
```
w5B[8] = w5B[7] /. z → ((# + I (1 - #)) &)  // ExpandAll
```

Out[87]=
$$-\frac{i\, k^{1/3}\, \text{Int}\left[(1-i)\, e^{-\frac{2 i k}{3}-2 k s^2}, \mathcal{D}_2\right]}{2\pi}$$

In[88]:=
```
w5B[9] = w5B[8] /. 𝒟2 → {s, -∞, ∞}
```

Out[88]=
$$-\frac{1}{2\pi} i\, k^{1/3}\, \text{Int}\left[(1-i)\, e^{-\frac{2 i k}{3}-2 k s^2}, \{s, -\infty, \infty\}\right]$$

In[89]:=
```
w5B[10] = w5B[9] /. Int → Integrate
```

Out[89]=
$$\text{ConditionalExpression}\left[-\frac{\left(\frac{1}{2}+\frac{i}{2}\right) e^{-\frac{2 i k}{3}}}{k^{1/6}\sqrt{2\pi}},\ \text{Re}[k] > 0\right]$$

In[90]:=
```
w5B[11] = Simplify[w5B[10], Assumptions → Re[k] > 0]
```

Out[90]=
$$-\frac{\left(\frac{1}{2}+\frac{i}{2}\right) e^{-\frac{2 i k}{3}}}{k^{1/6}\sqrt{2\pi}}$$

Here, I had to look at the FullForm of w5B[11] to get the pattern matching to work

In[91]:=
```
w5B[12] = w5B[11] /. Complex[Rational[-1, 2], Rational[-1, 2]] → e^(i π/4)/√2
```

Out[91]=
$$\frac{e^{-\frac{2 i k}{3}+\frac{i\pi}{4}}}{2 k^{1/6}\sqrt{\pi}}$$

Returning to the original parameter

In[92]:=
```
w5B[13] = w5B[12] /. k → Λ^(3/2) // PowerExpand
```

Out[92]=
$$\frac{e^{\frac{i\pi}{4}-\frac{2}{3} i \Lambda^{3/2}}}{2\sqrt{\pi}\,\Lambda^{1/4}}$$

5C Combining the contributions from the two saddle points

In[93]:=
```
w5C[1] = w5A[13] + w5B[13]
```

Out[93]=
$$\frac{e^{\frac{i\pi}{4}-\frac{2}{3} i \Lambda^{3/2}}}{2\sqrt{\pi}\,\Lambda^{1/4}}+\frac{e^{-\frac{i\pi}{4}+\frac{2}{3} i \Lambda^{3/2}}}{2\sqrt{\pi}\,\Lambda^{1/4}}$$

In[94]:=
```
w5C[2] = w5C[1] // ExpToTrig
```

Out[94]=
$$\frac{\text{Cos}\left[\frac{\pi}{4}-\frac{2\Lambda^{3/2}}{3}\right]}{\sqrt{\pi}\,\Lambda^{1/4}}$$

In[95]:=
```
w5C[3] = w5C[2] /. Λ → -λ
```

Out[95]=
$$\frac{\text{Cos}\left[\frac{\pi}{4}-\frac{2}{3}(-\lambda)^{3/2}\right]}{\sqrt{\pi}\,(-\lambda)^{1/4}}$$

Stating this result in standard asymptotic notation

$$A_i(\lambda) \sim \frac{\text{Cos}\left[\frac{\pi}{4}-\frac{2}{3}(-\lambda)^{3/2}\right]}{\sqrt{\pi}\,(-\lambda)^{1/4}} \text{ as } (\lambda \to -\infty)$$

I check this against Mathematica's implementation of the Airy function

In[96]:=
```
Module[{lab, standardArgs, F, FAsymptotic},
 F[λ_] := AiryAi[λ];
 FAsymptotic[λ_] := Cos[π/4 - 2/3 (-λ)^(3/2)] / (Sqrt[π] (-λ)^(1/4));

 lab = Stl@StringForm["Airy vs leading order asymptotic approximation"];
 standardArgs = StandardArgs["λ", "F", lab];
 Plot[{F[k], FAsymptotic[k]}, {k, -10, -0.1}, AxesLabel → {Stl["λ"], Stl["F"]},
  PlotLabel → lab, PlotStyle → {Black, Blue, Darker[Green, 0.5]},
  PlotLegends → Placed[{"AiryAi[λ]", "Cos[π/4 - 2/3 (-λ)^(3/2)] / (Sqrt[π] (-λ)^(1/4))"}, Right]]]
```

Out[96]=



Appendix A Detail of steepest descent curves for $\lambda < 0$

The equation for the steepest descent curves near the saddle point z = -i is

In[97]:=

$$A[1] = -y - x^2 y + \frac{y^3}{3} == \frac{2}{3}$$

Out[97]=

$$-y - x^2 y + \frac{y^3}{3} == \frac{2}{3}$$

I can solve this cubic

In[98]:=
```
A[2] = Solve[A[1], y] // Simplify
```

Out[98]=

$$\left\{\left\{y \to \left(1 + x^2 + \left(1 + \sqrt{-x^2 \left(3 + 3 x^2 + x^4\right)}\right)^{2/3}\right) \Big/ \left(1 + \sqrt{-x^2 \left(3 + 3 x^2 + x^4\right)}\right)^{1/3}\right\},\right.$$
$$\left\{y \to \left(i \left(-\left(-i + \sqrt{3}\right) \left(1 + x^2\right) + \left(i + \sqrt{3}\right) \left(1 + \sqrt{-x^2 \left(3 + 3 x^2 + x^4\right)}\right)^{2/3}\right)\right) \Big/ \left(2 \left(1 + \sqrt{-x^2 \left(3 + 3 x^2 + x^4\right)}\right)^{1/3}\right)\right\},$$
$$\left.\left\{y \to \left(9\, i \left(i + \sqrt{3}\right) \left(1 + x^2\right) - 9 \left(1 + i \sqrt{3}\right) \left(1 + \sqrt{-x^2 \left(3 + 3 x^2 + x^4\right)}\right)^{2/3}\right) \Big/ \left(18 \left(1 + \sqrt{-x^2 \left(3 + 3 x^2 + x^4\right)}\right)^{1/3}\right)\right\}\right\}$$

The solutions of interest are those that pass through the saddle point at z = -i, or for which y = -1 at x = 0

In[99]:=
```
A[3] = A[2] /. x → 0 // Simplify
```

Out[99]=

$$\{\{y \to 2\}, \{y \to -1\}, \{y \to -1\}\}$$

Discard the first solution

In[100]:=
```
A[4] = A[2][[2 ;; 3]] // Simplify
```

Out[100]=

$$\left\{\left\{y \to \left(i \left(-\left(-i + \sqrt{3}\right) \left(1 + x^2\right) + \left(i + \sqrt{3}\right) \left(1 + \sqrt{-x^2 \left(3 + 3 x^2 + x^4\right)}\right)^{2/3}\right)\right) \Big/ \left(2 \left(1 + \sqrt{-x^2 \left(3 + 3 x^2 + x^4\right)}\right)^{1/3}\right)\right\},\right.$$
$$\left.\left\{y \to \left(9\, i \left(i + \sqrt{3}\right) \left(1 + x^2\right) - 9 \left(1 + i \sqrt{3}\right) \left(1 + \sqrt{-x^2 \left(3 + 3 x^2 + x^4\right)}\right)^{2/3}\right) \Big/ \left(18 \left(1 + \sqrt{-x^2 \left(3 + 3 x^2 + x^4\right)}\right)^{1/3}\right)\right\}\right\}$$

Although not apparent from the form of the solution, these are real expressions

In[101]:=
```
LGrid[Table[{x, A[4][[1, 1, 2]], A[4][[2, 1, 2]]} , {x, -1, 1, 0.25}],
 "representative numerical values"]
```

Out[101]=

representative numerical values

-1.	$-2.2618 + 0.\,i$	$-0.339877 + 5.55112 \times 10^{-17}\,i$
-0.75	$-1.90764 - 1.11022 \times 10^{-16}\,i$	$-0.445534 + 0.\,i$
-0.5	$-1.5748 + 5.55112 \times 10^{-17}\,i$	$-0.587373 + 1.11022 \times 10^{-16}\,i$
-0.25	$-1.26984 - 8.32667 \times 10^{-17}\,i$	$-0.771536 + 1.38778 \times 10^{-16}\,i$
0.	$-1. + 0.\,i$	$-1. + 0.\,i$
0.25	$-1.26984 - 8.32667 \times 10^{-17}\,i$	$-0.771536 + 1.38778 \times 10^{-16}\,i$
0.5	$-1.5748 + 5.55112 \times 10^{-17}\,i$	$-0.587373 + 1.11022 \times 10^{-16}\,i$
0.75	$-1.90764 - 1.11022 \times 10^{-16}\,i$	$-0.445534 + 0.\,i$
1.	$-2.2618 + 0.\,i$	$-0.339877 + 5.55112 \times 10^{-17}\,i$

with the small imaginary parts arising from finite machine accurate arithmetic. Mathematica provide the *Chop* function for eliminating such contributions.

In[102]:=
```
LGrid[Table[{x, Chop@A[4][[1, 1, 2]], Chop@A[4][[2, 1, 2]]} , {x, -1, 1, 0.25}],
 "Representative numerical values"]
```

Out[102]=

Representative numerical values

−1.	−2.2618	−0.339877
−0.75	−1.90764	−0.445534
−0.5	−1.5748	−0.587373
−0.25	−1.26984	−0.771536
0.	−1.	−1.
0.25	−1.26984	−0.771536
0.5	−1.5748	−0.587373
0.75	−1.90764	−0.445534
1.	−2.2618	−0.339877

To represent the steepest descent path, one must switch branches. The steepest descent curve through $\mathcal{D}_1$ is the red branch for x < 0 but the green branch for x ≥ 0.

In[103]:=
```
Plot[{Chop@A[4][[1, 1, 2]], Chop@A[4][[2, 1, 2]]}, {x, -1, 1},
 PlotStyle → {Red, Green}, AxesLabel → {Stl["x"], Stl["y"]}]
```

Out[103]=



Mathematica has the *Piecewise* function for handling such situations

In[104]:=
```
Clear[SteepestDescentCurveSP1];
SteepestDescentCurveSP1[x_] :=
 Piecewise[{{(ⅈ (-(-ⅈ + √3) (1 + x^2) + (ⅈ + √3) (1 + √(-x^2 (3 + 3 x^2 + x^4)))^(2/3)))/
     (2 (1 + √(-x^2 (3 + 3 x^2 + x^4)))^(1/3)), x < 0},
   {(9 ⅈ (ⅈ + √3) (1 + x^2) - 9 (1 + ⅈ √3) (1 + √(-x^2 (3 + 3 x^2 + x^4)))^(2/3))/
     (18 (1 + √(-x^2 (3 + 3 x^2 + x^4)))^(1/3)), x ≥ 0}}]
```

But you still have to be careful. Note that a straight forward series expansion of the piecewise function gives the wrong answer.

In[106]:=
```
A[5] = {Normal@Series[SteepestDescentCurveSP1[x], {x, 0, 1}],
  Normal@Series[SteepestDescentCurveSP1[x], {x, 0, 1}, Assumptions → {x ≤ 0}]}
```

Out[106]=

$$\left\{\begin{cases} -1-x & x \le 0 \\ -1+x & \text{True} \end{cases}, \begin{cases} -1-x & x > 0 \\ -1+x & \text{True} \end{cases}\right\}$$

Finally, the approximate for for the steepest descent curve passing through z = -i is x-1.

In[107]:=
```
Plot[{Chop@SteepestDescentCurveSP1[x], x - 1} ,
 {x, -1, 1}, PlotStyle → {Black, Directive[Black, Dashed]}]
```

Out[107]=



Appendix: Graphics

In[16]:=

```
Module[{X = 3, Y = 3, Z = 5, δF = 0.005, image = 300, gSurface, lab, F, g},
 F[z_] := Exp[- z^3/3 + z];

 gSurface = Plot3D[Abs[F[x + I y]], {x, -X, X},
   {y, -Y, Y}, ImageSize → image, MeshFunctions -> {#3 &}, Mesh → 10,
   Boxed → False, AxesLabel → {Stl["x"], Stl["y"], Stl["|f(z)|"]},
   PlotRange → {{-X, X}, {-Y, Y}, {0, Z}},
   PlotLabel → TraditionalForm[F[z]]];
 g[1] = Show[{gSurface}];
 g[2] = ContourPlot[Abs[F[x + I y]], {x, -X, X}, {y, -Y, Y}, ImageSize → image,
   MeshFunctions -> {#3 &}, Mesh → 50 , PlotRange → {{-X, X}, {-Y, Y}, {0, Z}},
   FrameLabel → {{Stl[Rotate["y", -π/2]], ""}, {Stl["x"], "" }},
   PlotLegends → Automatic];

 Grid[{{g[1], g[2]}}]]
```

Out[16]=





In[21]:=

```
Module[
 {X = 3, Y = 3, Z = 5, δF = 0.005, image = 300, gSurface, C1, C2, C3, lab, F, g},
 F[z_] := Exp[- z^3/3 + z];

 C1 = {Directive[Yellow, Thick], Arrowheads[{0.0, 0.05, 0.05, 0.0}],
   Arrow[{{0, -Y}, {0, Y}}], {Black, Text["C1", {0, -2}]} };
 C2 = {Directive[Yellow, Thick], Arrowheads[{0.0, 0.05, 0.05, 0.0}],
   Arrow[{{0, -Y}, {0, 0}}], {Black, Text["C2", {0, -2}]} };
 C3 = {Directive[Yellow, Thick], Arrowheads[{0.0, 0.05, 0.05, 0.0}],
   Arrow[{{0, 0}, {X, 0}}], {Black, Text["C3", {2, 0}]} };
 g[1] = ContourPlot[Abs[F[x + I y]], {x, -X, X}, {y, -Y, Y}, ImageSize → image,
   MeshFunctions -> {#3 &}, Mesh → 50 , PlotRange → {{-X, X}, {-Y, Y}, {0, Z}},
   FrameLabel → {{ Stl[Rotate["y", -π/2]], ""}, {Stl["x"], "Contour for Airy" }},
   Epilog → {C1}];
 g[2] = ContourPlot[Abs[F[x + I y]], {x, -X, X}, {y, -Y, Y}, ImageSize → image,
   MeshFunctions -> {#3 &}, Mesh → 50 , PlotRange → {{-X, X}, {-Y, Y}, {0, Z}},
   FrameLabel → {{ Stl[Rotate["y", -π/2]], ""}, {Stl["x"], "Contour for BAiry" }},
   Epilog → {C2, C3}];

 Grid[{{g[1], g[2]}}]]
```

Out[21]=



In[37]:=

```
Module[{X = 3, Y = 3, Z = 5, δF = 0.005, image = 300,
  gSurface, saddlePoint, sdPath, ySD1, ySD2, C1, lab, F, g},
 F[z_] := Exp[- z^3/3 + z];
 saddlePoint[1] = {Green, PointSize[0.025],
   Point[{-1, 0}], {Black, Text["z_sp1", {-1, 0} + {-0.5, 0}]}};
 saddlePoint[2] = {Green, PointSize[0.025], Point[{1, 0}],
   {Black, Text["z_sp2", {1, 0} + {-0.5, 0}]}};
 ySD1[x_] := Sqrt[3] Sqrt[-1 + x^2];
 ySD2[x_] := -Sqrt[3] Sqrt[-1 + x^2];

 sdPath[1] = {GREEN, Arrowheads[{0.0, 0.05, 0.05, 0.0}],
   Arrow@Table[{x, ySD1[x]}, {x, -1, -2, -0.05}],
   Arrow@Table[{x, ySD2[x]}, {x, -2, -1, 0.05}],
   {Black, Text["D1", {-1.5, ySD2[-1.5]}]} };
 sdPath[2] = {GREEN, Arrowheads[{0.0, 0.05, 0.05, 0.0}],
   Arrow@Table[{x, ySD2[x]}, {x, 2, 1, -0.05}],
   Arrow@Table[{x, ySD1[x]}, {x, 1, 2, 0.05}],
   {Black, Text["D2", {1.5, ySD2[1.5]}]} };
 C1 = {Directive[Yellow, Thick], Arrowheads[{0.0, 0.05, 0.05, 0.0}],
   Arrow[{{0, -Y}, {0, Y}}], {Black, Text["C1", {0, -2}]} };

 g[1] = ContourPlot[Abs[F[x + I y]], {x, -X, X}, {y, -Y, Y}, ImageSize → image,
   MeshFunctions -> {#3 &}, Mesh → 50 , PlotRange → {{-X, X}, {-Y, Y}, {0, Z}},
   FrameLabel → {{Stl[Rotate["y", -π/2]], ""},
     {Stl["x"], "Original and deformed contour for Airy" }},
   Epilog → {saddlePoint[1], saddlePoint[2], sdPath[1], sdPath[2], C1},
   PlotLegends → Automatic]]
```

Out[37]=



In[38]:=

```
Module[{X = 3, Y = 3, Z = 5, δF = 0.005, image = 300, gSurface, saddlePoint,
  sdPath, ySD1, ySD2, C1, L1, L2, D1Approx, Ymax = 2.5, Xmax, lab, F, g},
 F[z_] := Exp[- z^3/3 + z];
 saddlePoint[1] = {Green, PointSize[0.025], Point[{-1, 0}]};
 ySD1[x_] := √3 √(-1 + x^2) ;
 ySD2[x_] := -√3 √(-1 + x^2) ;

 Xmax = Solve[ySD2[x] == -Ymax, x][[2, 1, 2]];
 sdPath = {GREEN, Arrowheads[{0.0, 0.05, 0.05, 0.0}],
   Arrow@Table[{x, ySD1[x]}, {x, -1, -Xmax, -0.05}],
   Arrow@Table[{x, ySD2[x]}, {x, -Xmax, -1, 0.05}],
   {Black, Text["D1", {-1.5, ySD2[-1.5]}]} };
 C1 = {Directive[Yellow, Thick], Arrowheads[{0.0, 0.05, 0.05, 0.0}],
   Arrow[{{0, -Y}, {0, Y}}], {Black, Text["C1", {0, -2}]} };
 L1 = {Directive[Green, Thick], Arrowheads[{0.0, 0.05, 0.05, 0.0}],
   Arrow[{{0, -Ymax}, {-Xmax , -Ymax}}], {Black, Text["L1", {-1, -Ymax}]},
   Arrow[{{-Xmax, Ymax}, {0 , Ymax}}], {Black, Text["L2", {-1, Ymax}]} };
 D1Approx = {Directive[Green, Thick, Dashed], Arrowheads[{0.0, 0.05, 0.05, 0.0}],
   Arrow[{{-1, -Ymax/2}, {-1 , Ymax/2}}],
   {Black, Text["D1 app", {-0.5, -Ymax / 4}]} };


 g[1] = ContourPlot[Abs[F[x + I y]], {x, -X, X}, {y, -Y, Y}, ImageSize → image,
   MeshFunctions -> {#3 &}, Mesh → 50 , PlotRange → {{-X, X}, {-Y, Y}, {0, Z}},
   FrameLabel → {{ Stl[Rotate["y", -π/2]], ""}, {Stl["x"], "Contour for Airy" }},
   Epilog → {saddlePoint[1], sdPath, C1,
     L1, D1Approx, Text["{0, -Ymax}", {1, -Ymax}]}]]
```

Out[38]=



In[65]:=

```
Module[{X = 3, Y = 3, Z = 5, δF = 0.005, image = 300,
  gSurface, saddlePoint, sdPath, xSD1, xSD2, C1, lab, F, g},
 F[z_] := Exp[- z^3/3 - z];
 saddlePoint[1] = {Green, PointSize[0.025],
   Point[{0, -1}], {Black, Text["z_sp1", {0, -1} + {-0.5, 0}]}};
 saddlePoint[2] = {Green, PointSize[0.025], Point[{0, 1}],
   {Black, Text["z_sp2", {0, 1} + {-0.5, 0}]}};
 xSD1[y_] := (Sqrt[-2 + y] (1 + y))/(Sqrt[3] Sqrt[y]);
 xSD2[y_] := -((-1 + y) Sqrt[2 + y])/(Sqrt[3] Sqrt[y]);

 sdPath[1] = {GREEN, Arrowheads[{0.0, 0.05, 0.05, 0.0}],
   Arrow@Table[{Chop@xSD1[y], y}, {y, -3, -0.05, 0.05}],
   {Black, Text["D1", {xSD1[-2], -2}]} };
 sdPath[2] = {GREEN, Arrowheads[{0.0, 0.05, 0.05, 0.0}],
   Arrow@Table[{Chop@xSD2[y], y}, {y, 0.05, 3.0, 0.05}],
   {Black, Text["D2", {xSD2[2], 2}]} };
 C1 = {Directive[Yellow, Thick], Arrowheads[{0.0, 0.05, 0.05, 0.0}],
   Arrow[{{0, -Y}, {0, Y}}], {Black, Text["C1", {0, -2}]} };

 g[1] = ContourPlot[Abs[F[x + I y]], {x, -X, X}, {y, -Y, Y}, ImageSize → image,
   MeshFunctions -> {#3 &}, Mesh → 50 , PlotRange → {{-X, X}, {-Y, Y}, {0, Z}},
   FrameLabel → {{ Stl[Rotate["y", -π/2]], ""},
     {Stl["x"], "Original and deformed contour for Airy with λ < 0" }},
   Epilog → {saddlePoint[1], saddlePoint[2], sdPath[1], sdPath[2], C1}]]
```

Out[65]=